SPSS Multiple Regression and Collinearity 
GET

  FILE='c:\temp\werner.sav'.

COMPUTE WTALB = WEIGHT + ALB.

EXECUTE.

/*Descriptives for Continuous Variables*/

DESCRIPTIVES

  VARIABLES=AGE HEIGHT WEIGHT PILL CHOL ALB CALCIUM ACID PAIR logwt agegroup

  hichol WTKG HTCM BMI HIBMI HIAGE CHOLCAT WTCAT WTALB

  /STATISTICS=MEAN STDDEV MIN MAX .

/*Frequencies for Categorical Variables*/

FREQUENCIES

  VARIABLES=AGE PILL

  /ORDER=  ANALYSIS .

Frequencies of categorical variables

/*Correlation Matrix of outcome variable and predictor variables*/

/*Pairwise Missing Values*/ 

CORRELATIONS

  /VARIABLES=CHOL AGE CALCIUM ACID ALB WEIGHT WTALB

  /PRINT=TWOTAIL NOSIG

  /MISSING=PAIRWISE .

/*Listwise Missing values*/

CORRELATIONS

  /VARIABLES=CHOL AGE CALCIUM ACID ALB WEIGHT WTALB

  /PRINT=TWOTAIL NOSIG

  /MISSING=LISTWISE .

/*Scatterplot Matrix*/

GRAPH

  /SCATTERPLOT(MATRIX)=CHOL AGE CALCIUM ACID ALB WEIGHT WTALB

  /MISSING=LISTWISE .

/*Multiple Linear Regression with Collinearity Diagnostics*/

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT CHOL

  /METHOD=ENTER AGE CALCIUM ACID ALB WEIGHT

  /SCATTERPLOT=(*SDRESID ,*ZPRED )

  /RESIDUALS HIST(ZRESID) NORM(ZRESID)

  /SAVE PRED COOK RESID SDRESID .

Regression with collinearity diagnostics

/*Descriptives on New Variables*/

DESCRIPTIVES

  VARIABLES=PRE_1 RES_1 SDR_1 COO_1

  /STATISTICS=MEAN STDDEV MIN MAX .

Descriptives

c:\temp\werner.sav

/*Compute a new variable that represents the current observation number*/

COMPUTE Observation = $CASENUM .

EXECUTE .

/*Scatter Diagram of the Cook's Distance vs. Observation Number*/

GRAPH

  /SCATTERPLOT(BIVAR)=Observation WITH COO_1

  /MISSING=LISTWISE .

/*Select Cases with Cook's Distance > .05*/

USE ALL.

COMPUTE filter_$=(COO_1 > .05).

VARIABLE LABEL filter_$ 'COO_1 > .05 (FILTER)'.

VALUE LABELS filter_$  0 'Not Selected' 1 'Selected'.

FORMAT filter_$ (f1.0).

FILTER BY filter_$.

EXECUTE .

/*List variables for the selected cases*/

LIST VARIABLES= id chol pre_1 res_1 sdr_1 coo_1.
ID   CHOL       PRE_1       RES_1       SDR_1       COO_1

 152  317   246.51175    70.48825     1.96537      .07079

2830  305   198.42170   106.57830     2.91152      .05432

3134  390   255.96029   134.03971     3.73160      .10521

Number of cases read:  3    Number of cases listed:  3

Regression model with perfect collinearity

Now, we fit a multiple regression model in which we deliberately include a variable, WTALB, which is perfectly collinear with weight and albumin (it is the sum of weight and albumin). SPSS detects this collinearity and produces a warning in the output. The output also doesn’t have information for all variables. 
/*Run a linear regression model with perfect collinearity*/ 

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT CHOL

  /METHOD=ENTER AGE CALCIUM ACID ALB WEIGHT WTALB

  /SCATTERPLOT=(*SDRESID ,*ZPRED )

  /RESIDUALS HIST(ZRESID) NORM(ZRESID)

  /SAVE PRED COOK RESID SDRESID .

Examine Another Data Set:

We now look at the Baseball dataset, which gives statistics for major league baseball players in 1986. This dataset, which contains information on players’ statistics for 1986 and for their entire careers, is provided as one of the SAS example datasets. We Fit a multiple linear regression model to predict their current salary based on a number of their current statistics, and examine collinearity. 
/*Open the SAS dataset baseball.sas7bdat*/

GET

  SAS DATA='C:\documents and settings\kwelch\sasdata2\baseball.sas7bdat'.

DATASET NAME DataSet2 WINDOW=FRONT.

/*Descriptive Statistics on the Baseball Data*/

DESCRIPTIVES

  VARIABLES=no_atbat no_hits no_home no_runs no_rbi no_bb yr_major cr_atbat

  cr_hits cr_home cr_runs cr_rbi cr_bb no_outs no_assts no_error salary

  /STATISTICS=MEAN STDDEV MIN MAX .

/*Pearson Correlations for the dependent variable and all predictors*/

CORRELATIONS

  /VARIABLES=salary no_atbat no_hits no_home no_runs no_rbi

  /PRINT=TWOTAIL NOSIG

  /MISSING=PAIRWISE .

/*Scatterplot Matrix*/

GRAPH

  /SCATTERPLOT(MATRIX)=salary no_atbat no_hits no_home no_runs no_rbi

  /MISSING=LISTWISE .

/*Regression with Collinear Predictors*/

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT salary

  /METHOD=ENTER no_atbat no_hits no_home no_runs no_rbi .

Regression with Collinear Predictors

/*Regression Deleting One Predictor*/

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT salary

  /METHOD=ENTER  no_hits no_home no_runs no_rbi   .

/*Regression Model with Only Two Predictors*/

REGRESSION

  /MISSING LISTWISE

  /STATISTICS COEFF OUTS R ANOVA COLLIN TOL

  /CRITERIA=PIN(.05) POUT(.10)

  /NOORIGIN

  /DEPENDENT salary

  /METHOD=ENTER  no_hits no_rbi   .

Regression Model with two Predictors

